Chapter 5: Practical Prolog Programming

- The Cut and Related Constructs
- Prolog vs. Pascal
- Definite Clause Grammars
Objectives

After completing this chapter, you should be able to:

- explain the effect of the cut.
- write Prolog programs for practical applications.
- use context-free grammars in Prolog.
Overview

1. The Cut and Related Constructs

2. Prolog vs. Pascal

3. Definite Clause Grammars
The Cut: Effect (1)

- The cut, written “!” in Prolog, removes alternatives that otherwise would have been tried during backtracking. E.g. consider this rule:
  \[ p(t_1, \ldots, t_k) :- A_1, \ldots, A_m, !, B_1, \ldots, B_n. \]
- Until the cut is executed, processing is as usual.
- When the cut is reached, all previous alternatives for this call to the predicate \( p \) are removed:
  - No other rule about \( p \) will be tried.
  - No other solutions to the literals \( A, \ldots, A_m \) will be considered.
The Cut: Effect (2)

- Example:

```
p(X) :- q(X), !, r(X).
p(X) :- s(X).
q(a).
q(b).
r(X).
s(c).
```

- With the cut, the query \( p(X) \) returns only \( X=a \).
- Without the cut, the solutions are \( X=a \), \( X=b \), \( X=c \).
- Exercise: Can the second rule about \( p \) ever be used?
The Cut: Effect (3)

Four-Port Model without Cut:

CALL → p(X) → q(X) → EXIT

FAIL → s(X) → r(X) → REDO
The Cut: Effect (4)

Four-Port Model with Cut:

CALL \rightarrow q(X) \rightarrow ! \rightarrow r(X) \rightarrow EXIT

FAIL \rightarrow s(X) \rightarrow REDO
The Cut: Effect (5)

• A call to the cut immediately succeeds (like true).
• Any try to redo the cut not only fails, but immediately fails the entire predicate call.
• In the SLD-tree, the cut “cuts away” all still open branches between
  ◊ the node where the cut was introduced (i.e. the child of which contains the cut), and
  ◊ the node where the cut is the selected literal.
The Cut: Effect (6)

\[ p(X) \]
\[ q(X), !, r(X) \]
\[ s(X) \]
\[ !, r(a) \]
\[ !, r(b) \]
\[ r(a) \]
\[ r(a) \]
The Cut: Effect (7)

- Before and after the cut, the backtracking is normal (only not through the cut):

\[
\begin{align*}
p(X,Y) & :\ q(X),
q2(X,Y). \\
p(X,Y) & :\ r1(X),
r2(X),
!, r3(X,Y),
r4(Y). \\
p(X,Y) & :\ s(X,Y).
q1(a).
q1(b). \quad q2(b,c).
r1(d).
r1(e). \quad r2(e).
r3(e,f).
r3(e,g). \quad r4(g).
s(h,i).
\end{align*}
\]

- The query \( p(X,Y) \) has solutions \( X=b, Y=c \) and \( X=e, Y=g \).
Cut: Improving Runtime (1)

- One application of the cut is to improve the runtime of a program by eliminating parts of the proof tree that cannot yield solutions or at least cannot yield any new solutions.

- Consider the predicate abs that computes the absolute value of a number:

  abs(X,X) :- X >= 0.
  abs(X,Y) :- X =< 0, Y is -X.

- When the first rule is successful, it is clear that the second rule does not have to be tried.
Cut: Improving Runtime (2)

- Consider now the goal \( p(X), \, \text{abs}(X,Y), \, Y > 5 \) with the facts \( p(3), \, p(0), \, p(-7) \).
- First \( p(X) \) succeeds with \( X \) bound to 3, then \( \text{abs}(3,Y) \) succeeds for \( Y=3 \), but then \( 3 > 5 \) fails.
- Now backtracking would normally first try to find an alternative solution for \( \text{abs}(3,Y) \), since there is another rule about \text{abs} that has not yet been tried.
- This is obviously useless, and the runtime can be improved by immediately backtracking to \( p(X) \).
Cut: Improving Runtime (3)

- With the cut, one can tell the Prolog system that when the first rule succeeds, the second rule cannot give anything new:

  \[
  \text{abs}(X, X) :- X \geq 0, !.
  \]
  \[
  \text{abs}(X, Y) :- X \leq 0, Y \text{ is } -X.
  \]

- Of course, one could have (should have) written the condition in the second rule \( X < 0 \).

- Then some (but not all) Prolog systems are able to discover themselves that the rules are mutually exclusive.
Cut: Improving Space (1)

- Making clear that a predicate has no other solution improves also the space (memory) efficiency.
- The Prolog system must keep a record ("choice-point") for each predicate call that is not yet complete (for backtracking into the predicate call later).
- Even worse, certain data structures within the Prolog system must be "frozen" when it is necessary to support later backtracking to this state.
- Then e.g. variable bindings must be logged (on the "trail") so that they can later be undone.
Cut: Improving Space (2)

• In imperative languages, when a procedure call returns, its stack frame (containing local variables and other information) can be reused.

• In Prolog, this is not always the case, because it might be necessary to reactivate the procedure call and search for another solution.

• E.g. consider the following program:

```prolog
p(X) :- q(X), r(X).
q(X) :- s(X), t(X).
s(a). s(b). t(a). t(b). r(b).
```
Cut: Improving Space (3)

- The call \( q(X) \) first exits with \( X=a \), but then \( r(a) \) fails, thus the call \( q(X) \) is entered again, which in turn reactives \( s(X) \).

  Upon backtracking, also the binding of \( x \) must be undone.

- In the above example, not much can be improved, because there really are alternative solutions.

- However, when a predicate call has only one solution, it should be executed like a procedure call in an imperative language.
Cut: Improving Space (4)

- Predicate calls that can have at most one solution are called **deterministic**.

  Sometimes one calls the predicate itself deterministic, but then one usually has a specific binding pattern in mind. E.g. `append` is deterministic for the binding pattern `bbf`, but it is not deterministic for `ffb`.

- For efficient execution, it is important that the Prolog system understands that a predicate call is deterministic. Here a cut can help.

  Actually, the cut in the definition of `abs` makes the predicate deterministic. In general, it might be important that `abs(0,X)` succeeds “two times”, Prolog is not allowed to automatically remove one solution. Deductive databases are set-oriented, there more powerful optimizers are possible.
Cut: Improving Space (5)

- Consider `abs` applied to a list:
  
  ```prolog
  abs_list([], []).
  abs_list([X|R], [Y|S]) :- abs(X, Y),
  abs_list(R, S).
  ```

- When the Prolog system thinks that `abs` is nondeterministic, it will keep the stackframe for each call to `abs` (and for the calls to `abs_list`).

- When a predicate calls a nondeterministic predicate, it automatically becomes nondeterministic, too.

  Only for the last body literal of the last rule about a predicate, the stack frame of the predicate is reused (under certain conditions), and thus does not remain, even when this body literal is non-deterministic.
Cut: Improving Space (6)

- In the above example, making `abs` deterministic (by means of a cut) is a big improvement.
- Then most Prolog systems will automatically deduce that also `abs_list` is deterministic.

  For the only possible binding patterns `bf` and `bb`.

- Usually, the outermost functor of the first argument is considered: Since it is “[]” for the first rule, and “.” for the second, always only one of the two rules is applicable (if the first argument is bound).
Cut: Improving Space (7)

- It is also possible to remove unnecessary stack frames at a later point.
- E.g. suppose that \( \texttt{abs} \) (and thus \( \texttt{abs_list} \)) remain nondeterministic, and consider the goal:
  \[
  \texttt{abs_list([-3,7,-4]}, \ X), \ !, \ \texttt{p(X)}.  
  \]
- The call to \( \texttt{abs_list} \) will leave many stack frames behind, but these are deleted by the cut.

It is probably better style to avoid the nondeterminism at the place where it occurs. However, one should not use too many cuts, and it might be easier to clean up the stack only at a few places.
The cut is also used to encode an “if then else”.

Consider the following predicate:

\[
\begin{align*}
p(X, Y) & :\neg q1(X), !, r1(X, Y). \\
p(X, Y) & :\neg q2(X), !, r2(X, Y). \\
p(X, Y) & : r3(X, Y).
\end{align*}
\]

This is equivalent to (assuming that \(q1\) and \(q2\) are deterministic):

\[
\begin{align*}
p(X, Y) & : q1(X), r1(X, Y). \\
p(X, Y) & : \neg q1(X), q2(X), r2(X, Y). \\
p(X, Y) & : \neg q1(X), \neg q2(X), r3(X, Y).
\end{align*}
\]
Cut: If-Then-Else (2)

- The formulation with the cut is a bit shorter. The difference becomes the bigger, the more cases there are.

- Furthermore, the runtime is shorter: In the version without the cut, \( q_1(X) \) is computed up to three times.

- But removing the cut in first version would completely change the semantics of the program. The cut is no longer only an “optimizer hint”.

Stefan Brass: Deductive Databases and Logic Programming
Univ. Halle, 2011
Cut: If-Then-Else (3)

- The logical semantics of programs with negation as failure ("\+") has been extensively studied and there are good proposals.
- I do not know of successful tries to give the cut a clear logical (declarative) semantics.

  The cut can basically be understood only operationally. One problem is that the cut is used for many different purposes, and it might be difficult to automatically discover for which one.

- Pure logic programmers try to avoid the cut, at least when it affects the logic of the program.
Cut: If-Then-Else (4)

- Prolog has an “if-then” operator \( \rightarrow \) that can be used to have the advantages of the cut, while making the logical intention clear.

- E.g. one could write the above procedure as

\[
p(X, Y) :- q1(X) \rightarrow r1(X,Y); \\
q2(X) \rightarrow r2(X,Y); \\
r3(X,Y).
\]

- \( A \rightarrow B \) has basically the same effect as \( A, !, B \).

However, if there should be further rules about \( p \), this cut does not remove the possibility to try these rules. It does remove alternative solutions for \( A \), and it does remove the possibility to try the disjunctive alternatives within the rule.
Conversely, one can implement negation as failure with the cut (\texttt{not} is only another name for \(\backslash +\)):

\[
\text{not}(A) :- \text{call}(A), !, \text{fail}.
\]
\[
\text{not}(_).
\]

The first rule ensures that if \(A\) succeeds, \text{not}(A) fails.

The second rule makes \text{not}(A) true in all other cases (i.e. when \(A\) fails).

Of course, if \(A\) should run into an infinite loop, also \text{not}(A) does not terminate.
Cut: One Solution (1)

- Suppose that email addresses of professors are stored as facts, and that the same person can have several email addresses:
  
  ```prolog
  prof_email(brass, 'sbrass@sis.pitt.edu').
  prof_email(brass, 'brass@acm.org').
  prof_email(spring, 'mspring@sis.pitt.edu').
  ...
  
  The cut can be used to select a single address of a given professor:
  
  ```prolog
  prof_email(brass, E), !, send_email(E).
  ```
Prolog has a built-in predicate `once` that can be used instead of the cut:

```prolog
once(prof_email(brass, E)), send_email(E).
```

`once` is defined as:

```prolog
once(A) :- call(A), !.
```

In the example, the following is equivalent:

```prolog
prof_email(brass, E) -> send_email(E).
```

However, the solution with `once` makes the intention clearer.
The cut can make programs wrong if predicates are called with unexpected binding patterns.

E.g. the predicate for the absolute value can also be written as follows (using the cut as in the if-then-else pattern):

\[
\text{abs}(X, X) :- X \geq 0, !. \\
\text{abs}(X, Y) :- Y \text{ is } -X.
\]

Since the second rule is executed only when the first rule is not applicable, it might seem that the test \( X =< 0 \) used earlier is superfluous.
**Cut: Dangers (2)**

- This is indeed true for the binding pattern `bf`, but consider now the call `abs(3,-3)`!
- In general, the rule is that the cut must be exactly at the point where it is clear that this is the right rule: Not too early and not too late.
- Here the unification must happen after the cut:
  
  ```prolog
  abs(X,Y) :- X >= 0, !, X = Y.
  abs(X,Y) :- Y is -X.
  ```
- This would work also with binding pattern `bb`.
Consider this predicate:

\[
\begin{align*}
\text{person}(X, \text{male}) & : \text{-} \text{man}(X), !. \\
\text{person}(X, \text{female}) & : \text{-} \text{woman}(X).
\end{align*}
\]

Since \text{man} and \text{woman} are disjoint, the cut was only added to improve the efficiency.

It works if \text{person} is called with binding pattern \textbf{bf} or \textbf{bb}. However, consider what happens if \text{person} is called with binding pattern \textbf{ff}!

It is interesting that here the more general binding pattern poses a problem, whereas in the \texttt{abs} example, the more specific binding pattern is not handled.
Types of Cuts

- Cuts in Prolog programs are usually classified into
  - *Green Cuts*: Do not modify the logical meaning of the program, only improve the runtime/space efficiency.
  - *Red cuts*: Modify the declarative meaning of the program.

Some authors also distinguish blue cuts: In this case, a good Prolog system should be able to determine itself that there are no further solutions. Blue cuts are intended only for very simple Prolog systems. “Grue Cuts”: Green or blue cuts.

Good Prolog programmers try to use red cuts only very seldom.
The cut is necessary for efficient Prolog programming, but it destroys the declarative meaning of the programs and can have unexpected consequences.

The better Prolog implementations get, the less important will be the cut.

Newer logic programming languages usually try to replace the cut by other constructs that have a more declarative meaning.

If possible, use `->`, `\+`, `once` instead.

Use the cut only as last resort.
Overview

1. The Cut and Related Constructs

2. Prolog vs. Pascal

3. Definite Clause Grammars
Prolog vs. Pascal

• “Prolog is different, but not that different.”
  This citation is probably from O’Keefe, The Craft of Prolog.

• In general, one can translate a given imperative algorithm (from Pascal, C, etc.) into Prolog.
  The resulting program might be not the best possible program for the task, just as a word-by-word translation from e.g. German to English gives bad English. But at least, if one knows how to solve a problem in an imperative language, one should also be able to write a Prolog program for it.

• The goal of this section is to teach some typical patterns of Prolog programming.
## Data Types (1)

<table>
<thead>
<tr>
<th>Pascal</th>
<th>Prolog</th>
</tr>
</thead>
<tbody>
<tr>
<td>integer</td>
<td>integer</td>
</tr>
<tr>
<td>real</td>
<td>float</td>
</tr>
<tr>
<td>char</td>
<td>ASCII-code (integer)</td>
</tr>
<tr>
<td></td>
<td>atom</td>
</tr>
<tr>
<td>string</td>
<td>list of ASCII-codes</td>
</tr>
<tr>
<td></td>
<td>atom</td>
</tr>
<tr>
<td></td>
<td>string (in some Prologs)</td>
</tr>
<tr>
<td>file</td>
<td>stream</td>
</tr>
<tr>
<td></td>
<td>atom (alias, in some Prologs)</td>
</tr>
<tr>
<td></td>
<td>switching standard IO</td>
</tr>
</tbody>
</table>
## Data Types (2)

<table>
<thead>
<tr>
<th>Pascal</th>
<th>Prolog</th>
</tr>
</thead>
<tbody>
<tr>
<td>enumeration type</td>
<td>set of atoms</td>
</tr>
<tr>
<td>(variant) record</td>
<td>composed term</td>
</tr>
<tr>
<td>(union/struct in C)</td>
<td>functor(field1, ..., fieldN)</td>
</tr>
<tr>
<td>array</td>
<td>list</td>
</tr>
<tr>
<td></td>
<td>set of facts: a(i, valI)</td>
</tr>
<tr>
<td></td>
<td>term: a(val1, ..., valN)</td>
</tr>
<tr>
<td>pointer</td>
<td>structured terms (e.g., lists)</td>
</tr>
<tr>
<td></td>
<td>otherwise like array index</td>
</tr>
<tr>
<td></td>
<td>partial data structures (terms with variables)</td>
</tr>
</tbody>
</table>
5. Practical Prolog Programming

Variables (1)

- One can assign a value to a Prolog variable only once.
  
  This is the biggest difference to imperative programming.

- Afterwards it is automatically replaced everywhere by its value.
  
  I.e. it ceases to exist as a variable.

- Thus, there is no possibility to assign a different value during normal, forward execution.

  Of course, with backtracking, one can go back to the point where the variable was still unbound. But then all other variable bindings that happen since that point in time are also undone.
Thus, one uses a different variable for every value:

```prolog
procedure p(n: integer, var m: integer);
begin
  m := n * 2;
  m := m + 5;
  m := m * m
end
```

This is an artificial example. Normally, one would compute the value of \( m \) in a single expression.
Using a new variable for every assignment is obviously possible for sequential/linear code.

Loops are formulated in Prolog by recursion, thus one can also get a fresh set of variables for every iteration (see Slide 5-53 for more efficient solution):

- for i := 1 to n do writeln(i);
- loop(N) :- loop_body(1, N).
  loop_body(I, N) :- I > N.
  loop_body(I, N) :- I =< N, write(I), nl,
                   Next_I is I + 1,
                   loop_body(Next_I, N).
Variables (4)

- For variables passed between procedures (in/out-parameters), a Pascal variable is split into two Prolog variables: One for the input value, and one for the output value ("accumulator pair").

- `procedure double(var n: integer);
  begin
    n := n * 2
  end
- `double(N_In, N_Out) :-
  N_Out is N_In * 2.`
Variables (5)

- Global variables in Pascal should be made predicate parameters in Prolog.
  
  Values that are passed unchanged from predicate to predicate are called “context arguments”.

- If there are too many global variables, one can pack them into a structure (composed term) which can be passed as a unit.
  
  One should declare a predicate for each variable to get/set the value in the structure.
• One can represent a global variable also with a fact in the dynamic database:
  □ \( x := x + 1 \)
  □ \( x(X), \text{retract}(x(X)), !, X1 \text{ is } X + 1, \text{assert}(x(X1)) \).

  There is always only one fact about the predicate \( x \) which contains the current value of \( x \).

• Some Prolog systems have support for destructive assignments and global variables, but that is very system-dependent.

  In GNU-Prolog, there is, e.g., \texttt{g_assign/2, g_read/2, g_array_size/2}. In SWI-Prolog, see \texttt{flag/3, setarg/3}. In Sepia/ECLiPSe, see \texttt{setval/2}. 
Conditions: If, Case (1)

• Example:

```
procedure min(i, j: integer; var m: integer);
begin if i < j then m := i else m := j end
```

• There are basically three possibilities to translate conditional statements:
  ◦ One rule per case, body contains the complete condition (i.e. the negation of all previous cases):
    
    ```
    min(I, J, M) :- I < J, M = I.
    min(I, J, M) :- \+ (I < J), M = J.
    
    Of course, one would write I >= J instead of \+ (I < J). But not all conditions can be inverted so simply.
    ```
Conditions: If, Case (2)

- Translations of if-statements, continued:
  - A rule per case, each starts with the “else if”-condition and a cut:
    \[
    \text{min}(I, J, M) :\text{-} \quad I < J,!, \quad M = I.
    \]
    \[
    \text{min}(I, J, M) :\text{-} \quad M = J.
    \]
  - Using the conditional operators, i.e. a rule with a body of the form (Cond→ Then ; Else):
    \[
    \text{min}(I, J, M) :\text{-} \quad (I < J \rightarrow M = I; \ M = J).
    \]
  - If possible, it is best to express the condition in the head of the rule (see next slide).
Conditions: If, Case (3)

- Example (condition in the rule head):
  - procedure `p(c: color; var i: integer);`  
  ```
  begin
    case c of
      red: i := 1;
      green: i := 2;
      blue: i := 3;
    end
  end
  ```
  - `p(red, I) :- I = 1.`
  - `p(green, I) :- I = 2.`
  - `p(blue, I) :- I = 3.`
  
  Of course, one would simplify this further to, e.g., `p(red, 1).`
The first solution (complete condition in every rule) is logically cleanest, it also permits to understand each rule in isolation.

However:

◊ There is a certain duplication of code.

◊ In many Prolog systems, it will leave a choice point behind if the first alternative is chosen.

Some Prolog systems are intelligent enough to understand that \( I < J \) and \( I \geq J \) exclude each other, thus no choice point is needed. The solution with the cut or \( \rightarrow \) does not have this problem.
Conditions: If, Case (5)

- In general, it is better to use the explicit conditional operator \( \rightarrow \) instead of the cut \( ! \), because this makes the purpose of the cut clear.
  - However, this makes the syntactical structure of the rules more complicated.

- Most Prolog systems have an index (hash table) over the outermost functor of the first argument.
  - Thus, if it is possible to code the condition in the rule head (first argument), this will be especially efficient, and no choicepoint will be generated.
Loops: Tail-Recursion (1)

- Loops are usually written as end-recursions.
- One should try to make sure that only a constant amount of memory is needed, not an amount linear in the number of executions of the loop body.
- A Prolog system reuses the memory of a rule invocation when the last body literal is called and there are no more alternatives.

If necessary, one can use a cut to make clear that other rules are not applicable. It is best when the recursive call is the last literal of the last rule about the predicate.
Loops: Tail-Recursion (2)

- Example (list length):

```prolog
procedure length(l: list; var n: integer);
begin
    n := 0;
    while l <> nil do begin
        n := n + 1;
        l := l ↑.next
    end
end
```

Stefan Brass: Deductive Databases and Logic Programming
Univ. Halle, 2011
Loops: Tail-Recursion (3)

- Direct Translation to Prolog:

  ```prolog
  length(L, N) :-
  length(L, 0, N).

  length(L, N_In, N_Out) :-
  L = [], !,
  N_Out = N_In.

  length(L, N_In, N_Out) :-
  N_Next is N_In + 1,
  [_|L_Next] = L,
  length(L_Next, N_Next, N_Out).
  ```
Loops: Tail-Recursion (4)

• Using the rule head for the conditions (makes the cut unnecessary):

\[
\text{length}([\ ], \text{N\_In}, \text{N\_Out}) :- \\
\quad \text{N\_Out} = \text{N\_In}.
\]

\[
\text{length}([\_|L\_Next], \text{N\_In}, \text{N\_Out}) :- \\
\quad \text{N\_Next is N\_In + 1}, \\
\quad \text{length}(L\_Next, \text{N\_Next}, \text{N\_Out}).
\]
Loops: Tail-Recursion (5)

• Alternative (elegant, but not tail recursive):

\[
\text{length}([], 0).
\]

\[
\text{length}([\_|L\_Rest], N) :-
\]

\[
\text{length}(L\_Rest, N\_Rest),
\]

\[
N \text{ is } N\_Rest + 1.
\]

• In this case, the system must return to a rule invocation after the recursive call.

• Thus, many systems will need memory that is linear in the length of the list. But efficiency is not all!

The memory will become free after the call to the length predicate. If the lists are not extremely long, the more elegant solution should be preferred.
Loops: Tail-Recursion (6)

- Consider again the for-loop:
  - `for i := 1 to n do writeln(i);`
  - `loop(N) :- loop_body(1, N).`
    `loop_body(I, N) :- I > N.`
    `loop_body(I, N) :- I =< N, write(I), nl, Next_I is I + 1, loop_body(Next_I, N).`
- The last call (for `I = N+1`) probably leaves a choice point behind.
- But then the stack frames of all recursive calls are protected, and the memory complexity is linear.
Loops: Tail-Recursion (7)

- In this case, it might be better to use a cut to make clear that the second rule is no alternative when the first rule is applicable:

  \[
  \text{loop\_body}(I, N) \leftarrow I > N, !.
  \]

  \[
  \text{loop\_body}(I, N) \leftarrow \text{write}(I), \text{nl},
  \text{Next}_I \text{ is } I + 1,
  \text{loop\_body}(\text{Next}_I, N).
  \]

- In this case, one has only the choice between two solutions that are both not nice.

  A good Prolog system might discover that the two conditions \(I > N\) and \(I \leq N\) are mutually exclusive. Then there is no problem.
Loops: Backtracking (1)

• Some loops can also be written with “repeat” and backtracking.

• This is only possible if only the side effect of the loop body is important (input/output, changes to the dynamic database), but no variable bindings must be kept from one iteration to the next.

• If it is possible, it is very efficient.

Not only the stack frames of predicate invocations are reused, but also all term structures that were built up. This is important for simple Prolog systems that have no garbage collection: There, space on the heap is recycled only upon backtracking. For loops that run an indefinite amount of time (command loops etc.), this should be used.
Loops: Backtracking (2)

- **procedure** `skip(c: character);`
  `var c1: character;`
  `begin`
  `  repeat`
  `    read(c1)`
  `  until c1 = c`
  `end`

- `skip(C) :- repeat, get_code(C1), C1 = C, !.`

The cut is important to avoid the re-execution of the loop if later in the program something fails and backtracking starts.
• Another typical pattern for a backtracking loop is to iterate over all solutions to a predicate:

```prolog
print_all_composers :-
    composer(FirstName, LastName),
    write(LastName),
    write(’, ’),
    write(FirstName),
    nl,
    fail.
print_all_composers.
```

The fact at the end ensures that the predicate ultimately succeeds.
Overview

1. The Cut and Related Constructs
2. Prolog vs. Pascal
3. Definite Clause Grammars
Definite Clause Grammars (1)

- In Prolog, one can directly write down grammar rules (of context-free grammars).
  
  Actually, the grammar formalism is even more expressive, since one can include arbitrary Prolog code as an additional condition for the applicability of a grammar rule.

- A simple preprocessor translates the grammar rules into standard Prolog rules.

- Thus, Prolog has a tool like yacc/bison built-in.
  
  yacc/bison are standard tools used in compiler construction: Given a context-free grammar (with certain restrictions), they produce a C program that checks the syntactic correctness of the input. One can extend the grammar with program code for processing the input.
Definite Clause Grammars (2)

- However, the goal of the Prolog grammar formalism is not compiler construction, but natural language processing (e.g., machine translation):
  - There one needs more complicated grammars.
  - E.g., non-deterministic grammars are possible in Prolog, but not in compiler-construction tools.

In natural language (e.g., English), there are ambiguous words, phrases, and sentences. These can easily be processed with backtracking in Prolog. In programming languages (e.g., C, Prolog), the meaning of every construct must be completely clear.

- However, the efficiency requirements are not so strong, since the inputs are usually not very long.
Definite Clause Grammars (3)

- Comparison with yacc/bison, continued:
  - In Prolog, arbitrary context-free grammars are possible, in yacc/bison only LALR(1) grammars.
    The condition in compiler construction tools ensures that efficient parsing is possible: The decisions for building the parse tree are done backtrack-free with only a single token lookahead.
  - In Prolog, grammar rules can be mixed with arbitrary program code. This can contain additional checks for selecting a grammar rule.
    In yacc/bison, one can include C code that is executed when a grammar rule is applied. In this way, one can, e.g. generate output. However, the C code cannot influence the parsing decisions, e.g. choose one of several possible grammar rules.
Definite Clause Grammars (4)

- The grammar on the next slide describes the commands of a simple text adventure game:
  - Such games are similar to books, in which the reader can give the main actor commands and influence in this way the storyline.
    
    Therefore, they are called “interactive fiction”. Titles like “Zork” by Infocom were very popular.

- In principle, the user can input any English sentence. In practice, most commands
  - move the player around in the adventure world
  - apply objects found in certain locations.
Definite Clause Grammars (5)

```
command --> verb, noun_phrase.
command --> [go], direction.
command --> direction.
command --> [quit].
direction --> [north].
direction --> [south].
direction --> [east].
direction --> [west].
verb --> [take].
verb --> [examine].
noun_phrase --> noun.
noun_phrase --> [the], noun.
noun --> [key].
noun --> [lamp].
```
Definite Clause Grammars (6)

- Nonterminal grammar symbols (syntactic categories) are written like standard Prolog predicates.
- Terminal symbols (expected input tokens) are written in [...].
- The syntactic analysis is done with the predicate “phrase”, e.g.
  - phrase(command, [take, the, lamp]). → yes.
  - phrase(command, [lamp, the, north]). → no.

Of course, one needs more than “yes/no”. This is done by attributes of the grammar symbols (predicate arguments). See below.
Implementation (1)

- In principle, every nonterminal symbol $N$ is translated into a Prolog predicate that is true for all lists of input tokens that are derivable from $N$.

- A naive solution (not the Prolog solution) would generate rules like

  ```prolog
  command(X) :- append(Y, Z, X),
                 verb(Y), noun_phrase(Z).
  command([go|X]) :- direction(X).
  command(X) :- direction(X).
  command([quit]).
  ```
The above solution is too inefficient: Especially the arbitrary splitting of the input list with `append` causes a lot of unnecessary backtracking.

The real implementation of grammar rules uses a data structure called “difference lists”:

- E.g. the list `\[a, b, c\]` is represented by a pair of lists `\[a, b, c | X\]` and `X`.
- A special case is the pair `\[a, b, c, d, e\], [d, e]`:

  This also represents the list `\[a, b, c\]`.
Thus, every nonterminal symbol is translated into a predicate with two arguments:

- Input list (total rest of input tokens before the nonterminal symbol is processed).
- Output list (rest of input after the nonterminal).

The difference between both lists are the input symbols derivable from the nonterminal symbol:

```prolog
command(X, Z) :- verb(X, Y), noun_phrase(Y, Z).
command(X, Z) :- X = [go|Y], direction(Y, Z).
command(X, Z) :- direction(X, Z).
command(X, Z) :- X = [quit|Z].
```
Implementation (4)

- I.e. every predicate cuts off from the input list the prefix it can process, and hands the rest to the next predicate.
- The syntax analysis is then done by calling the predicate for the start symbol of the grammar with the complete input list and the empty list as the rest:
  
  \[
  \text{command([take, the lamp], [])}.
  \]
- Thus, the predicate `phrase` is defined as:
  \[
  \text{phrase(Start, Input) :- Goal =.. [Start,Input,[]], call(Goal).}
  \]
Attributes (1)

- Usually, it is not sufficient to know that the input is syntactically correct, but one needs to collect data from the input.
- Therefore, the nonterminal symbols can have arguments (which correspond to attributes in attribute grammars).
- The preprocessor for grammar rules simply extends the given literals by two further arguments for the input and output token lists.

The given arguments are left untouched.
command(V,O) --> verb(V), noun_phrase(O).
command(go,D) --> [go], direction(D).
command(go,D) --> direction(D).
command(quit,nil) --> [quit].
direction(n) --> [north].
direction(s) --> [south].
direction(e) --> [east].
direction(w) --> [west].
verb(take) --> [take].
verb(examine) --> [examine].
noun_phrase(O) --> noun(O).
noun_phrase(O) --> [the], noun(O).
noun(key) --> [key].
noun(lamp) --> [lamp].
Further Possibilities (1)

- One can include arbitrary Prolog code in the syntax rules.
- It must be written in {...} in order to protect it from the rewriting done by the preprocessor.
- E.g. it might be easier to store a list of game objects as facts, and to use only a single grammar rule for nouns:

  \[
  \text{noun}(O) \rightarrow [O], \{\text{object}(O, _, _)\}.
  \]

  The additional arguments of \text{object} could be the initial location and the description of the object.
Further Possibilities (2)

- The cut `!`, the disjunction (which can also be written `|`), and the if-then symbol `→` do not need to be included in `{...}`.

  One can also use parentheses `(…)` to structure the alternatives.

- For instance, the optional article before the noun can also be encoded in a single rule:

  `noun_phrase(O) --> ([the] | []), noun(O).`

- The cut can help to improve the efficiency of the syntax analysis.
Further Possibilities (3)

- The left hand side of the syntax rule can contain a “look-ahead terminal”, e.g.

\[ p, [a] \rightarrow q, [a]. \]

means that the production \( p \rightarrow q \) can only be applied if \( a \) is the next token.

This is translated to \( p(X_1,X_4) :- q(X_1,X_2), X_2=[a|X_3], X_4=[a|X_3] \), i.e. the look-ahead terminal” is inserted back into the input stream after the rule is processed. In the example, \( X_2 = X_4 \), thus the \( a \) is not consumed.
Efficiency Improvements

- Avoid left recursion.
  This is usually not only inefficient, but wrong: At least for incorrect inputs it easily gets into an endless recursion.

- Think about possible cuts, especially before tail recursions.

- It might be possible to use the Prolog index over the first argument.

```prolog
lookahead(Token), [Token] --> [Token].
stmt --> lookahead(Token), stmt(Token).
stmt(if) --> [if], cond, [then], stmt.
...
stmt(id) --> [id], [':='], expression.
```
Formal Syntax

\[
\begin{align*}
g\_rule & \rightarrow g\_head, [\'--\'>], g\_alt. \\
g\_head & \rightarrow non\_terminal, ([','], terminal | []). \\
g\_alt & \rightarrow g\_if, (['|'], g\_alt | []). \\
g\_if & \rightarrow g\_rhs, (['-\'>], g\_rhs | []). \\
g\_rhs & \rightarrow g\_item, ([,], g\_rhs | []). \\
g\_item & \rightarrow terminal. \\
g\_item & \rightarrow non\_terminal. \\
g\_item & \rightarrow variable. \\
g\_item & \rightarrow [!'']. \\
g\_item & \rightarrow ['](], g\_alt, [')']. \\
g\_item & \rightarrow ['{']}, prolog\_goal, ['}']. \\
non\_terminal & \rightarrow any\_callable\_prolog\_term. \\
terminal & \rightarrow ['['], (toks | []), ['']]. \\
toks & \rightarrow any\_prolog\_term, ([','], toks | []).
\end{align*}
\]
Lexical Analysis (1)

- The input to the syntax analysis (parser) is usually a list of word symbols, called tokens.
- Of course, one could also use a list of characters (atoms or ASCII codes).
- However, since the combination of characters to words is simple, a more efficient algorithm (without backtracking) can be used.
- This reduces a long list of characters to a short list of words. Then the more complex algorithm can work on a shorter input.
Lexical Analysis (2)

• The module that is responsible for transforming a sequence of characters into a sequence of word symbols (lexical analysis) is called the scanner.

• The separation of lexical analysis and syntax analysis is also useful because the scanner can suppress:
  ◊ white space between tokens
    Usually, any sequence of spaces, tabulator characters, and line ends is permitted.
  ◊ comments.

• This simplifies the syntax analysis.
Lexical Analysis (3)

- The following example program reads input characters until the line end.
- It skips spaces and composes sequences of letters to words. Other characters (punctuation marks etc.) are treated as one-character tokens.
- The main work is done by a predicate `scan` that gets the current input character as first argument:

```
scanner(TokList) :-
    get_char(C),
    scan(C, TokList).
```
Lexical Analysis (4)

scan(’\n’, []) :- !.
scan(’ ’, TokList) :- !,
    get_char(C),
    scan(C, TokList).
scan(C, [Word|TokList]) :-
    letter(C), !,
    read_word(C, Letters, NextC),
    name(Word, Letters),
    scan(NextC, TokList).
scan(C, [Sym|TokList]) :-
    name(Sym, [C]),
    get_char(NextC),
    scan(NextC, TokList).
Lexical Analysis (5)

- The predicate `read_word` reads a list of letters starting with character `C`. It returns the character `NextC` that follows after the word:

  ```prolog
  read_word(C, [C|MoreC], NextC) :-
  letter(C), !,
  get_char(C2),
  read_word(C2, MoreC, NextC).
  read_word(C, [], C).
  ```

- The predicate `letter` defines which characters can appear in words:

  ```prolog
  letter('a').
  ...
  ```